

§7. Арифметические действия над элементами массива

Элементы массива в программах используются так же, как и обычные переменные. Над ними можно выполнять операции, допустимые для переменных соответствующего типа. Элементам типа *integer* можно присваивать результат выполнения операций $+$, $-$, $*$, **div**, **mod** (участвовать в качестве операндов они могут и в операции $/$). Для элементов типа *real* возможны операции $+$, $-$, $*$, $/$.

Рассмотрим примеры выполнения арифметических действий над элементами одномерных числовых массивов.

Пример 1. В массиве хранятся данные о количестве осадков (в мм), выпадавших ежемесячно на протяжении года. Составить программу, которая вычисляет суммарное количество осадков за год, а также среднемесячное количество осадков.

Массив данных о количестве осадков за 12 месяцев зададим в виде целочисленных констант:

```
const h: array[1..12] of integer=(43, 40, 42, 46, 61, 82, 90, 81, 60, 50, 53, 50);
```

Вычисленную сумму будем помещать в переменную *Sum* целого типа, а среднее значение - в переменную *Sred* вещественного типа.

Последовательность действий при вычислении суммы и среднего арифметического такова:

- Присваиваем начальное значение суммы $Sum := 0$.
- Вычисление суммы производим в цикле *for*. Значение параметра (счетчика) цикла *i* изменяется от 1 до 12. На каждом шаге цикла к значению переменной *Sum* прибавляем значение очередного элемента (в этом примере количество осадков за очередной месяц) $Sum := Sum + h[i]$.
- По завершению цикла вычисляем среднее значение $Sred := Sum / 12$ и выводим результаты.

Программа имеет вид:

```
program Primer7_1;  
const h: array[1..12] of integer=(43, 40, 42, 46, 61, 82, 90, 81, 60, 50, 53, 50);  
var Sum, i: integer; Sred: real;  
begin
```

```

Sum:=0; { начальное значение суммы }
for i:=1 to 12 do Sum := Sum +h[i]; { вычисление суммы }
Sred := Sum/12; { вычисление среднего }
writeln('суммарное количество = ',Sum, ' среднемесячное = ', Sred:5:2);
end.

```

Протестируем программу. В окне вывода получим:

```
суммарное количество = 698 среднемесячное = 58.17
```

Алгоритм вычисления произведения элементов числового массива отличается лишь тем, что начальное значение $P:=1$. На каждом шаге цикла выполняется умножение на значение очередного элемента $P:=P * A[i]$.

В некоторых задачах требуется выполнять арифметические действия не над всеми элементами массива, а лишь удовлетворяющими определенным условиям, например, положительными или отрицательными, четными или нечетными числами, значениями, кратными заданному числу.

Пример 2. Составить программу, которая формирует массив из 12 случайных целых чисел, принадлежащих промежутку $[-40; 40]$, и вычисляет сумму положительных чисел.

Сформируем массив случайных чисел на указанном промежутке.

Вычисленную сумму будем помещать в переменную **S**. Присвоим начальное значение $S:=0$.

Суммирование производится в цикле *for* при выполнении условия $A[i]>0$.

Программа будет выглядеть так:

```

program Primer7_2;
var A: array[1..12] of integer;
    i, S: integer;
begin
    for i:=1 to 12 do A[i]:=random(81) - 40; { формирование массива }
    S:=0;
    for i:=1 to 12 do
        if A[i]>0 then S:=S+A[i]; { вычисление суммы положительных чисел }
    for i:=1 to 12 do write(A[i]:4); { вывод чисел в строку через 4 позиции }
    writeln; { переход на новую строку }
    writeln('сумма положительных чисел = ',S); { вывод суммы }
end.

```

Результат работы может выглядеть так:

```
-21 -7 35 -9 10 36 22 -19 1 -34 15 37
сумма положительных чисел = 156
```

Аналогичным способом выполняются арифметические действия над элементами массива, значения которых удовлетворяют, например, условиям:

```

if A[i] mod 2 = 0 then...      { четные числа }
if A[i] mod 2 = 1 then...     { нечетные числа }
if A[i] mod N = 0 then..     { числа, кратные N }.....
    
```

Пример 3. Составить программу, которая формирует массив из 7 случайных целых чисел, принадлежащих промежутку [1; 5], и вычисляет произведение нечетных чисел.

Сформируем массив случайных чисел на указанном промежутке.

Вычисленное произведение будем помещать в переменную *P*. Начальное значение *P:=1*.

Будем умножать *P* на значение очередного элемента только на тех шагах цикла *for*, где выполняется условие *A[i] mod 2 = 1*.

Программа будет выглядеть так:

```

program Primer7_3;
var A: array[1..7] of integer;
    i, P: integer;
begin
    for i:=1 to 7 do A[i]:=random(5)+1;           { формирование массива }
    P:=1;
    for i:=1 to 7 do
        if A[i] mod 2 = 1 then P:= P * A[i];     { вычисление произведения }
    for i:=1 to 7 do write(A[i]:4);             { вывод чисел в строку через 4 позиции }
    writeln;
    writeln('произведение нечетных чисел = ',P); { вывод произведения }
end.
    
```

Результат работы может выглядеть так:

2	1	4	5	4	5	3
произведение нечетных чисел = 75						

Пример 4. В массиве *A* хранятся данные о расходе топлива на 100 км пробега автомобилей восьми моделей. При замене топлива на новую марку расход уменьшается на 10%. Составить программу, которая вычисляет расход нового топлива и выводит в таблицу номер модели автомобиля, расход старого и нового топлива с двумя десятичными цифрами.

Пусть данные о расходе топлива хранятся в массиве констант вещественного типа: *const A: array[1..8] of real = (7.4, 7.0, 6.8, 5.9, 6.4, 9.1, 7.6, 8.2);*

Расход нового топлива будем вычислять непосредственно в процедуре выводе *writeln*. Это позволит обойтись без создания второго массива. Для помещения данных в таблицу используем форматированный вывод.

Программа может выглядеть так:

```

program Primer7_4;
const A: array[1..8] of real = (7.4, 7.0, 6.8, 5.9, 6.4, 9.1, 7.6, 8.2);
    var B: array[1..8] of real;
        i: integer;
begin
    writeln('модель расход-стар расход-нов');           { заголовок таблицы }
    for i:=1 to 8 do
        writeln(i:4, A[i]:10:2, 0.9*A[i]:10:2);       { вычисление и вывод }
end.
    
```

Результат работы будет выглядеть так:

модель	расход-стар	расход-нов
1	7.40	6.66
2	7.00	6.30
3	6.80	6.12
4	5.90	5.31
5	6.40	5.76
6	9.10	8.19
7	7.60	6.84
8	8.20	7.38

- ?
1. Какие операции допустимы для элементов массива целых чисел?
 2. Какие операции применимы для элементов массива вещественных чисел?

Упражнения

1. Составьте программу, которая вычисляет среднюю плотность раствора (в кг/м³) по результатам измерения плотности 7 проб (задаются в массиве констант: 1205, 1350, 1170, 1320, 1220, 1310, 1180).

2. Составьте программу, которая осуществляет ввод в массив с клавиатуры размеров 6 файлов в байтах (от 100 до 2000) и вычисляет:

- а) суммарный размер файлов в килобайтах;
- б) средний размер файла.

3. Составьте программу, которая формирует массив из 40 случайных целых чисел от 0 до 20, и производит одно из следующих вычислений:

- а) среднее арифметическое нечетных чисел;
- б) сумму чисел, не превышающих 10;
- в) сумму чисел, кратных 3.

На экран выводятся исходные значения и результаты вычислений..